

Themenübersicht September 2002

Ausgabe: 9 / 2002

- Nice to know
- Benchmark: ANSYS 6.1 auf INTEL Itanium2 unter WinXP 64-bit
- Zyklisch symmetrische Modalanalyse
- Magnetischer Dipol vor leitender Kugel
- Zylindrischer Kontakt mit linearen und quadratischen Elementen
- Beispiel zur neuen Fluid-Strukturkopplung in ANSYS

- Unter anderem in der nächsten Ausgabe:

Beitrag zu Stabilitätsuntersuchungen

Selektives MODOPT bei PSD

Beispiel für die Kopplung von elektrischem Feld und Wärmeübertragung

In eigener Sache:

Die Zusendung dieser Informationen erfolgt ausschließlich auf Wunsch des Empfängers und kann jederzeit unter www.cadfem.de beendet werden.

Wenngleich die vorliegenden Informationen mit größter Sorgfalt erstellt worden sind, weisen wir darauf hin, dass die Verwendung dieser unter Ausschluss jeglicher Gewährleistung erfolgt.

Nice to know

Ausgabe: 9 / 2002

ANSYS

- Die Result-Übersicht kann mit der Befehlsfolge /OUTPUT,Dateiname und SET,LIST,2 in eine Datei umgeleitet werden.
- Der Export von IGES-Files sollte ausschliesslich über FILE>EXPORT erfolgen. Beim Export mit der CDWRITE-Funktion können weitere Daten, die nicht zur Geometriebeschreibung gehören ungewollt mitexportiert werden.
- Im /PSF-Befehl, der die Darstellung der Flächenlastrandbedingungen steuert, sind Optionen enthalten, die über das Menü nicht erreichbar sind. Z.B. /PSF,PRES,TANX,2 liefert eine Darstellung des in tangentialer x-Richtung aufgebrauchten Drucks.
- Um nur Elemente an der Oberfläche einer Struktur zu selektieren reicht es bei einem Hexaeder oder Quadnetz aus die Oberflächenknoten zu selektieren und die daran hängenden Elemente auszuwählen. Bei Tetraedernetzen werden dabei aber auch die Elemente selektiert, die nur mit einer Ecke die Oberfläche berühren. Abhilfe schafft hier das Aufbringen einer Dummymlast mit SFA auf die gewünschte Oberfläche. Nachdem die Belastung auf das Netz übertragen wurde (SBCTAN) kann mit dem ESEL,S,SFE,PRES,Last die Elementlage selektiert werden.
- Im /POST26 werden mit dem STORE-Befehl Daten zu Variablen abgelegt. STORE,NEW öffnet eine neues Datenfeld. STORE,MERGE überschreibt die Daten bereits bestehender Variablen. STORE,APPEN hängt neue Zeiten an bestehende Variablen an. Der STORE- Befehl ist in Print und Plotbefehlen bereits enthalten.
- Für Linkelemente ist es nicht direkt möglich, die Elementspannungen mit dem *GET-Befehl abzufragen. Abhilfe schafft hier der Umweg über ein Elementtable:

```
ETAB,T_1,LS,1  
*GET,S_X,ELEM,elementnr,ETAB,T_1
```

DesignSpace

- Bei der Bauteilübergabe von DesignSpace 6.1 nach Pro/E 2000i2 oder Pro/E 2001 wird die "Aktive CAD-Dateie" zweimal angezeigt. Nur einer der beiden Links funktioniert wirklich.

Zum Beheben dieses Fehlers öffnen Sie bitte die Registry (Start/Ausführen) und geben Sie "regedit" ein. Danach wechseln Sie in folgendes Verzeichnis:

```
HKEY_LOCAL_MACHINE\SOFTWARE\ANSYS, Inc.\CAD Integration\PlugIns
```

In den Fällen, in denen das Problem mit dem "doppelten Link" auftritt, finden Sie in diesem Ordner die Unterordner "ProEngineer" und "Pro/Engineer."

Wenn Sie den Ordner "Pro/Engineer" löschen, funktioniert das Pro/E-PlugIn zu DS 6.1 wieder einwandfrei.

Benchmark: ANSYS 6.1 auf INTEL Itanium2 unter WinXP 64-bit

Ausgabe: 9 / 2002


ANSYS unterstützt jetzt auch den neuen INTEL Itanium2 unter WinXP 64-bit (Windows.NET Advanced Server RC1 (build 3668)). Damit ist eine effektive Nutzung des Parallel Processing möglich.

Der beschriebene Benchmark wurde gerechnet mit freundlicher Unterstützung von:

Intel Solution Service, Schweden und

TÜV Nord, Bereich Energie- und Systemtechnik, Hamburg

Als Testmodell wurde ein Gussdeckel mit ca. 90.000 SOLID92 Elementen (ca. 420.000 DOF) vernetzt. Für das Material wurde ein bilineares Werkstoffgesetz (BISO) verwendet. Das Rechenmodell berücksichtigt grosse Verformungen.



Rechnerdaten:

Itanium2/Box/Chipset: (Alpha/Beta/Prod) - Itanium2 Vehicle - Based on EPIC architecture

4way Itanium2 900Mhz L3 3MB Cache

Chipset E8870 (6.4 GB/s bandwidth)

8GB Memory (DDR PC2600) (Max 64GB Memory in the machine)

SCSI UW320 Controller

1 x 18GB DISK UW-SCSI


System Bus: 400Mhz, 128-bit wide, 6.4 GB/s bandwidth

Chipset: Intel E8870 Chipset



Benchmark: ANSYS 6.1 auf INTEL Itanium2 unter WinXP 64-bit

Ausgabe: 9 / 2002

Der Benchmark wurde mit dem PCG-, dem Sparse- und dem AMG-Solver gefahren. Erwartungsgemäß ergab sich, dass der PCG wegen der Nichtlinearität des Problems nur mäßig geeignet ist. Bemerkenswert ist aber die hohe Skalierung des PCG-Solvers (Faktor 1.8 auf 2 CPUs, 3.2 auf 4 CPUs)



Der für das Problem besser geeignete Sparse-Solver erreicht auf 2 CPUs ebenfalls eine gute Skalierung (Faktor 1.4).



Der AMG-Solver schliesslich löste das Problem sogar schneller als der Sparse-Solver und besitzt zudem einen hohen Skalierungsgrad (Faktor 1.7 auf 2 CPUs, Faktor 2.7 auf 4 CPUs).

Da der Sparse-Solver ab etwa 500.000 DOF seine hohe Performance verliert, bietet sich zur Simulation grosser nichtlinearer Modelle der AMG-Solver an und ist somit eine sinnvolle Erweiterung des Solverspektrums von ANSYS. Die Lizenz für den AMG-Solver kann zu Testzwecken über eCADFEM bezogen werden.

Zylindrischer Kontakt mit linearen und quadratischen Elementen

Ausgabe: 9 / 2002

Problem:

Es gibt verschiedene Richtlinien wie die Target und Kontaktfläche zu wählen ist. Bei zylindrischem Kontakt (bei konzentrischen Kontaktflächen) ist es darüberhinaus wichtig auch darauf zu achten, dass sich die Elementierung der beiden Flächen entspricht. Alternativ dazu können auch quadratische Elemente verwendet werden. Bei Verwendung von nichtlinearem Material kann es bei Nichtbeachtung dieser Richtlinie zu unsinnigen plastischen Verformungen kommen.

Erläuterung:

In der Online-Hilfe findet man diverse Hinweise nach welche Kriterien die Target- und Kontaktflächen auszuwählen sind. Unter anderm gilt, dass die feiner vernetzte Fläche die Kontaktseite sein sollte.

Bei zylindrischem Kontakt ergibt sich aber noch eine Besonderheit:

Elemente mit linearem Ansatz müssen mit gleicher Elementierung auf den beiden Flächen verwendet werden.

Werden Elemente mit quadratischem Ansatz gewählt können diese dagegen auch mit unterschiedlichem Netz eingesetzt werden.

Der Grund für dieses Verhalten ist, dass bei einem linearen Ansatz die Integrationspunkte (an denen der Kontakt defaultmäßig abgefragt wird) auf Geraden liegen. Konzentrische Kontaktflächen werden daher ebenfalls mit Geraden abgebildet. Die Ecken schneiden sich in die gegenüberliegenden Geraden. Dies führt zu physikalisch unbegründeten Spannungskonzentrationen.

Bei gleicher Elementierung ergibt sich das Problem der einschneidenden Kanten nicht. Dieses Vorgehen resultiert in den besten Ergebnissen.

Quadratische Elemente können durch den quadratische Ansatz die Krümmung besser abbilden, wodurch diese Elemente auch bei ungleicher Elementierung noch akzeptable Ergebnisse liefern.

Beispiel:


Die Inputfiles stellen ein Viertelmodell (2D) zweier Kreisinge dar, die gegeneinander gepresst werden.

Im ersten Inputfile werden lineare Elemente verwendet. Mit dem Schalter i kann zwischen gleicher und unterschiedlicher Vernetzung der beiden Ringe gewählt werden.

Im zweiten Inputfile werden Elemente mit quadratischem Ansatz verwendet.

Die Auswertung der von Mises Vergleichsspannung zeigt sehr deutlich, wie bei den linearen Elementen die unterschiedliche Vernetzung zu Spannungsspitzen führt.

Die quadratischen Elemente schliesslich ergeben auch bei unterschiedlichem Netz gute Resultate und eignen sich daher vor allem für Fälle, in denen die Target- und Contact- Flächen sich



Zylindrischer Kontakt mit linearen und quadratischen Elementen

Ausgabe: 9 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
fini
/clear
! i=1 -> gleiches Netz
! i=2 -> unterschiedliches Netz
i=1

/prep7
et,1,182 ! lineare Elemente
et,2,169
et,3,171
mp,ex,1,2e5

pcir,1,2,0,90
pcir,2,3,0,90
esiz,,3

*if,i,eq,1,then
lesi,1,,,8
*else
lesi,1,,,6
*endif
lesi,7,,,8
mshkey,1
ames,all

lsel,s,,,7
nsl,s,1
type,2
esurf
lsel,s,,,1
nsl,s,1
type,3
esurf

/solu
nset,s,loc,x,0
d,all,ux
nset,s,loc,y,0
d,all,uy
csys,1
nset,s,loc,x,0,1
sf,all,pres,1
alls
solve

/post1
plesol,s,eqv
```

```
fini
/clear

/prep7
et,1,183 ! quadratische Elemente
et,2,169
et,3,172
mp,ex,1,2e5

pcir,1,2,0,90
pcir,2,3,0,90

esiz,,3
lesi,1,,,6
lesi,7,,,8
mshkey,1
ames,all

lsel,s,,,7
nsl,s,1
type,2
esurf
lsel,s,,,1
nsl,s,1
type,3
esurf

/solu
nset,s,loc,x,0
d,all,ux
nset,s,loc,y,0
d,all,uy
csys,1
nset,s,loc,x,0,1
sf,all,pres,1
alls
solve

/post1
plesol,s,eqv
```

Zyklisch symmetrische Modalanalyse

Ausgabe: 9 / 2002

Problem:

Vor allem bei zyklisch symmetrischen Modellen bietet es sich an durch Ausnutzung der Symmetrie Rechenzeit einzusparen. Seit ANSYS 6.0 ist hierfür ein Satz von Befehlen implementiert, der die erforderlichen Aktionen zur Berücksichtigung der zyklischen Symmetrie in benutzerfreundlicher Form zur Verfügung stellt.

Erläuterung:

Mit dem neuen Eingabesatz kann in ANSYS auf sehr einfache Weise ein zyklisch symmetrisches Modell berechnet und die Ergebnisse am Vollmodell ausgewertet werden. Es reicht aus einen Sektor, der sich ganzzahlig bis zum Vollkörper von 360° wiederholen muss, zu modellieren. Der Sektor kann in einem beliebigen zylindrischen Koordinatensystem erstellt werden.

Im Preprozessor wird ANSYS mit dem Befehl CYCLIC mitgeteilt, dass es sich um eine zyklische Analyse handelt. Dabei sind lineare und geometrisch nichtlineare statische Analysen mit symmetrischen Belastungen und Modalanalysen mit und ohne Vorspannung (auch aus einer geometrisch nichtlinearen Berechnung) möglich.

Es ist zwar zulässig an der sog. unteren und oberen Kante des modellierten Sektors unterschiedliche Netze zu generieren. Jedoch liefern übereinstimmende Netze bessere Ergebnisse. Ist CYCLIC vor dem Vernetzen eingegeben worden, wird mit den Befehlen AMESH und VMESH automatisch ein übereinstimmendes Netz an den beiden Randkanten generiert.

Im Solutionteil kann mit dem /CYCOPT Befehl der Lösungsvorgang gesteuert werden.



Für das Postprocessing stellt ANSYS die Befehle EXPAND, /CYCEXPAND und CYCPHASE zur Verfügung.

Mit dem /CYCEXPAND werden die Ergebnisse expandiert, ohnedass die Datenbasis verändert wird. Damit bleibt der Speicherbedarf klein, aber es können nur die Originalknoten für weitere Berechnungen angewählt werden. Mit dem EXPAND Befehl werden nicht nur die Ergebnisse expandiert, sondern auch alle übrigen virtuellen Knoten generiert. Dies führt zu sehr grossen Files, die aber wie in einer nicht zyklischen Analyse in vollem Umfang zum weiteren Postprocessing herangezogen werden können.

Mit CYCPHASE schliesslich kann der Phasenwinkel berechnet werden, bei dem bestimmte Grössen Ihr Maximum oder Minimum erreichen. Dieser Phasenwinkel kann dann an das /CYCEXPAND übergeben werden.

Beispiel:

Im Beispiel wird für eine runde Platte mit Rippen eine nicht vorgespannte Modalanalyse durchgeführt. Die Berechnung erfolgt an 1/12 des Modell (30°).



Zyklisch symmetrische Modalanalyse

Ausgabe: 9 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
fini
/clear

/prep7
et,1,181
r,1,.3

mp,ex,1,210000
mp,prxy,1,0.3
mp,dens,1,7.8e-8

csys,1

k,1,1,0,0
k,2,10,0,0
k,3,1,30,0
k,4,10,30,0
k,5,1,15,0
k,6,10,15,0
k,7,1,15,2
k,8,10,15,4

a,1,2,4,3
a,5,6,8,7
aglu,all

lsel,s,,,1,3,2
lsel,a,,,5,7,2
lesl,all,,,8

cyclic
mshkey,1
amesh,all

/solu
antype,modal
modopt,lanb,2
cycopt,status
solve

/post1
set,list
/cycexpand,,on
set,3,1
pldisp,2
```

Magnetischer Dipol vor leitender Kugel

Ausgabe: 9 / 2002

Problem:

Es sollen magnetische Flussdichte und die induzierte Stromdichte in eine elektrisch leitende Kugel vor einem magnetischen Dipol berechnet werden.

Erläuterung:




Mit ANSYS können dreidimensionale elektromagnetische Felder im Nahfeld mit Elementen des Typs SOLID117 berechnet werden. Dabei werden Effekte wie Wirbelströme/Stromverdrängung berücksichtigt; dazu ist in den leitenden Regionen die Keyoption 1 auf 1 zu setzen.

Die durchgeführte harmonische Berechnung ergibt für den Lastfall zwei Ergebnis-Sets: eins für den Real- und eins für den Imaginärteil der Lösung. Der Imaginärteil entspricht einem Phasenversatz von 90° , so dass dieser zur Auswertung der induzierten Ströme herangezogen werden muss (SET,1,1,,1).

Bei der Verwendung elektromagnetischer Elementtypen empfiehlt sich die Vernetzung mit Hexaeder-Elementen im Wirbelstrom-Bereich. Diese können durch Anwenden des VSWEEP Befehls auch für kugelförmige Körper einfach erzeugt werden.

Beispiel:

Es wird eine stromführende Leiterschleife vor einer leitenden Kugel modelliert. Diese wird mit einer Einheitsstromdichte von 1A/m^2 beaufschlagt und eine harmonische Berechnung bei 10 MHz durchgeführt. Die sich einstellende magnetische Flussdichte und Stromdichte werden als Imaginärteil des Lösungssets dargestellt.



Magnetischer Dipol vor leitender Kugel

Ausgabe: 9 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
/prep7
frq=1e7
curr=1
lslope=1e-3
dslope=.1e-3
rsph=lslope
dsph=3*lslope
et,1,117,0 ! magnetic region
et,3,117,1 ! harmonic -> eddy current region
mp,murx,1,1
mp,murx,2,1
mp,murx,3,1
mp,rsvx,3,1/56e5
bloc,-0.5*(lslope+dslope)*0,0.5*(lslope+dslope),-
0.0*dslope,0.5*dslope,-
0.5*(lslope+dslope),0.5*(lslope+dslope)
bloc,-0.5*(lslope-dslope)*0,0.5*(lslope-dslope),-
0.0*dslope,0.5*dslope,-0.5*(lslope-dslope),0.5*(lslope-
dslope)
vsbv,1,2
sphere,10*lslope,,0, 90
wpoffs,0,0,dsph
sphere,rsph,,0,90 ! sphere
sphere,.1*rsph,,0,90 ! inner sphere
sphere,1.3*rsph,,0,90 ! air region around sphere
vovl,all

.....

mat,1
type,1
vsweep,all
cm,schleife,elem
esel,none
vsel,s,loc,z,dsph-rsph,dsph+rsph ! sphere
lsel,s,loc,z,-1.3*rsph+dsph,-rsph+dsph
asll,s
vsla,u
vsel,u,loc,x,0,.1*rsph
esize,dslope
mat,3
type,3
vsweep,all
vsel,s,loc,z,dsph,dsph+rsph !center of sphere
vsel,r,loc,x,0,.1*rsph
esize,dslope
mshape,1,3D
mshkey,0
mat,3
type,3 ! Eddy Current Region
vmesh,all
cm,körper,elem
```

```
vsymm,x,all
vsymm,y,all
nummrg,node,1e-5,1e-5
nummrg,kp,1.5e-5,1.5e-5

/solu
esel,s,mat,,1 ! attach current density to elements
nsl,s
nsl,r,loc,x,.5*(lslope-dslope),.5*(lslope+dslope)
esln,r,1
bfe,all,js,,curr

.....

alls
csys,2
nsl,s,loc,x,20*lslope ! External
nodes
d,all,az,0
csys,0
allsel
antype,harmic !harmonic analysis
harfrq,frq !frequency for
analysis
eqslv,sparse !sparse solver
solve

/post1
set,1,1,,1 ! imaginary part -> phase=90°
esel,u,mat,,2
/view,,-1,,-1,-1
plns,b,sum
plvect,b,,,,vect,elem,1
plvect,jt,,,,vect,elem,1
```

Den vollständigen Einagbestrom finden Sie wie immer auf unserer Homepage.

Beispiel zur neuen Fluid-Struktur Kopplung in ANSYS

Ausgabe: 9 / 2002

Problem:

ANSYS bietet seit Rev.6.1 eine Reihe von neuen Möglichkeiten eine gekoppelte Fluid-Struktur-Berechnung sehr einfach aufzusetzen und zu steuern. Diese neuen Features werden hier vorgestellt.

Erläuterung:

Bei einer dynamischen Fluid-Struktur-Kopplung muß zunächst einmal das reine strömungs- und strukturmechanische Berechnungsproblem gelöst werden. Hinzukommen der Lasttransfer zwischen Fluid und Festkörper, die Netznachführung bei großer Deformation des Festkörpers sowie die Kontrolle des gesamten Lösungsalgorithmus.



Das Aufsetzen einer solchen Berechnung erfordert im wesentlichen folgende Schritte:

1. Vernetzung des Festkörpers und des Fluids mit den jeweils geeigneten Elementen. Dabei werden beide Gebiete völlig unabhängig voneinander vernetzt, die Netze hängen also nicht zusammen.
2. Aufbringen der struktur- und strömungsmechanischen Randbedingungen und dazugehöriger Lösungseinstellungen. Auch hier wird jedes der Gebiete völlig getrennt von einander behandelt.
3. Die Knoten, an denen die Fluid-Struktur-Kopplung stattfindet, werden durch ein fsin-Label gekennzeichnet (surface loads, sfx-Kommandos).
4. Als letztes werden die Einstellungen zur Kontrolle des Lösungsalgorithmus getroffen. Dieses sind insbesondere der Zeitschritt, bei dem eine Interaktion zwischen Fluid und Festkörper erfolgen soll, die maximale Anzahl der Iterationen pro Zeitschritt (stagger) und das Konvergenzkriterium für jeden Zeitschritt.

Der Lasttransfer erfolgt an dem mit dem fsin-Label gekennzeichneten Knoten völlig automatisch und erfordert keine weiteren Eingaben.

Beispiel:

Untersucht wird die Deformation einer dünnen Metallblechs unter einer transienten Druckbelastung. Am Eintritt in den Strömungskanal wird mit Hilfe des function editors eine sinusförmige Druckbelastung aufgegeben, die die Strömung antreibt und damit eine Verformung der Metallblechs bewirkt. Das berechnete Zeitintervall beträgt 1 Sekunde, bei einer Zeitschrittlänge von 0.002 Sekunden. 5 Stagger (Iterationen) pro Zeitschritt sind ausreichend um für jeden Zeitschritt eine konvergierte Lösung zu erzielen.



Beispiel zur neuen Fluid-Struktur Kopplung in ANSYS

Ausgabe: 9 / 2002

ANSYS Eingabesatz (ANSYS 6.1):

```
/filna,blech
...
/prep7
!*** Parameter ***
a=1.0
b=2.0
...
!***Sinusfunktion fuer Druck***
freq=1          !1 Hz
omega=360*freq    !2*Pi*freq in Grad
amp=20000        !Ampitude (Druck)
*DIM,ptab,TABLE,6,4,1
! Begin of equation: amp*sin(omega*{TIME})
ptab(0,0,1)= 0.0, -999
ptab(2,0,1)= 0.0
ptab(3,0,1)= amp
ptab(4,0,1)= omega
ptab(5,0,1)= 0.0
ptab(6,0,1)= 0.0
ptab(0,1,1)= 1.0, -1, 0, 1, 18, 3, 1
ptab(0,2,1)= 0.0, -1, 9, 1, -1, 0, 0
ptab(0,3,1)= 1, -2, 0, 1, 17, 3, -1
ptab(0,4,1)= 0.0, 99, 0, 1, -2, 0, 0
! End of equation: amp*sin(omega*{TIME})
!*** Aufbau des Modells ***
/PREP7
et,1,141
et,2,82
keyopt,1,4,1          !enable disp DOFs
rect,0,gap,0,b/2-thick
rect,0,gap,b/2-thick,b/2+thick
!*** Merge areas without lever
...
!*** Attributes ****
asel,u,,,2
aatt,,1      !Typ=1
asel,s,,,2
aatt,,2      !Typ=2
alls
!*** Meshing ***
...
!*** Boundary Conditions ***
!*** exterior walls
lsel,s,,,4,12,8
nsl,s,1
d,all,vx,0
d,all,vy,0
!*** symmetry
lsel,s,,,14,16,2
lsel,a,,,19
nsl,s,1
d,all,vx,0
```

```
!*** Fix pressure at lower left corner
lsel,s,,,1,13,12
nsl,s,1
d,all,pres,%ptab%
!*** outlet
lsel,s,,,11,20,9
nsl,s,1
d,all,pres,0
!*** fsi interfaces
!structural side
lsel,s,,,5,7,1
nsl,s,1
sf,all,fsin,1
!fluid side
lsel,s,,,3,9,6
lsel,a,,,18
nsl,s,1
sf,all,fsin,1
!*** Structural displacements at lever
...
/solu
flda,solu,flow,1
flda,solu,ale,1
flda,solu,turb,1
flda,solu,tran,t

flda,prot,dens,air-si
flda,prot,visc,air-si

mp,ex,1,0.7e11
mp,nuxy,1,0.3
mp,dens,1,7800

fsan,1      !fsi on
fsou,1      !write every
fsit,5      !maximum 5 stagger loops
fsti,1      !end time
fsdt,0.002 !fsi delta time
fsor,solid  !solve structure first
fstr,solid,tran !transient fluid
fstr,fluid,tran !transient structural
fsin,cons
deltim,0.002 !structural time step
fsre,all,0.2 !relax load transfer
fsc0,all,1e-3
flda,time,step,0.002 !fluid time step
flda,time,glob,10 !global iterations
alls
solve
```

Den vollständigen Eingabestrom finden Sie auf unserer Homepage.

Termine rund um CADFEM

Ausgabe: 9 / 2002

Aktuelle Seminartermine:

- ANSYS 6.0/6.1 Update
16.09.02 in Leinfelden-Echterdingen bei Stuttgart
23.09.02 in Burgdorf bei Hannover
21.10.02 in Grafing bei München
- ANSYS Viskoelastizität und Viskoplastizität
10.-11.10.02 in Burgdorf bei Hannover
- Infotag: DesignXplorer+AGP
14.10.02 in Leinfelden-Echterdingen bei Stuttgart
15.10.02 in Burgdorf bei Hannover
17.10.02 in Grafing bei München
- Einführung in die Optimierung
15.-16.10.02 in Leinfelden-Echterdingen bei Stuttgart
- Materialmodelle in LS-DYNA
17.10.02 in Leinfelden-Echterdingen bei Stuttgart
- Kontaktberechnungen mit LS-DYNA
18.10.02 in Leinfelden-Echterdingen bei Stuttgart
- ANSYS Strukturdynamik
22.-25.10.02 in Grafing bei München
- Einführung in die Ermüdungsfestigkeit
22.-23.10.02 in Leinfelden-Echterdingen bei Stuttgart
- Fußgängerschutz-Simulation mit LS-DYNA
25.-26.09.02 in Liederbach bei Frankfurt
24.-25.10.02 in Leinfelden-Echterdingen bei Stuttgart
- ANSYS Faserverbundwerkstoffe
22.-23.10.02 in Burgdorf bei Hannover
- ANSYS Elektrodynamik 2D
24.-25.10.02 in Burgdorf bei Hannover
- Infotag ANSYS/FEM in der Mechatronik
28.10.02 in Grafing bei München
- Optislang
28.-31.10.02 in Leinfelden-Echterdingen bei Stuttgart
- Schnuppertraining: Konstruktionsbegleitend berechnen – DesignSpace
29.10.02 in Grafing bei München
- Explizite Dynamik mit ANSYS/LS-DYNA
29.-31.10.02 in Burgdorf bei Hannover
- Elektromagnetik-Struktur-Kopplung
30.-31.10.02 in Grafing bei München

Termine rund um CADFEM

Ausgabe: 9 / 2002

Weitere Veranstaltungen:

- **An event to remember:**

- 20. CADFEM Users` Meeting 2002
International Congress on FEM Technology
October 9-11, 2002
Kultur- und Congress-Centrum "Graf-Zeppelin-Haus"
Friedrichshafen, Germany
- Messe MATERIALICA "World of Product Engineering"
30.09. - 02.10. 2002, Neue Messe München
- VDI Tagung Berechnung und Simulation im Fahrzeugbau
01.10.-02.10.2002 in Würzburg (Maritim)
- 11. Aachener Kolloquium Fahrzeug- und Motorentechnik
07.10 - 09.10.2002 in Aachen
- 2. CESUM - Central European SolidWorks Users Meeting
07.10 - 08.10.2002 in Wiesbaden